
State Machines

This lecture defines state machines and describes how to document them and how to implement them using Verilog.
After this lecture you should be able to: design a state machine based on an informal description of its operation, document
it using state transition diagrams and tables, write a synthesizable Verilog description of it and convert between these three
descriptions. Revision 3: corrections and clarifications.

Ɪntroduction

The state of a register is its value. A state machine is
a description of how the state changes in response to
inputs.
The schematic of a state machine is a register

whose next value is selected by a combination of in-
puts and the current state:

State Machine Descriptions

State machines can be described by state-transition
tables or state-transition diagrams.
A state transition table has columns for the current

state, the input value(s), and the corresponding next
state.
The example below is for a motor controller with

one bit of state and two inputs: one starts the motor
and one stops it. Themotor stops if both are asserted:

current
state

inputs next
statestart stop

0 0 0 0
0 0 1 0
0 1 0 1
0 1 1 0
1 0 0 1
1 0 1 0
1 1 0 1
1 1 1 0

A state machine can also be described using a state
transition diagram. Each state is represented by a cir-
cle labelled with the state. Lines with arrows show

transitions between states and the input conditions
needed for that transition. The state transition dia-
gram for this state machine would be:

0 1

start == 1 && stop == 0

 start == 0 && stop == 1
|| start == 1 && stop == 1

State transitions, which are changes in the register
value, happen at the rising edge of the clock.
State transition diagramsmay omit transitions that

don’t result in a change of state (e.g. from 0 to 0 or
from 1 to 1).
Exercise 1: Draw the schematic for this state machine. Write an
always_ff statement that implements this state machine.

State Machine Outputs

We can generate outputs that depend on the state.
For aMoore state machine the outputs are a func-

tion of the current state (only). For aMealy state ma-
chine1 the outputs are also a function of the current
inputs:

We can often choose state encodings where the
state is also the output. Otherwise, the output for
each state can be documented as another column in
the state transition table or added to each circle in the
state transition diagram.

1We will not cover Mealy state machines in this course.

Ɪnteracting State Machines

Circuits often contain multiple state machines. The
state of one can be an input to another.
For example, a multi-digit BCD counter may

contain several single-digit counters. The value
of a counter would increment when the next-less-
significant digit value was 9.
Another example would be the traffic light con-

troller example below. One register represents which
lights are turned on. Another register is the number
of seconds remaining before the transition to the next
light state.

Sequence Generators

Sequence generators are state machines that produce
a desired sequence of values. Typical inputs may in-
clude those to restart the sequence (typically called
reset), pause or continue the sequence (hold or en‑
able), or change the values generated (e.g. up/down,
increment, ormaximum value inputs).

Counter

The state transition state diagram and for a two-bit
counter with a reset input is:

00 01 10 11

r:  reset == 1

reset==0 reset==0 reset==0

r r r

A simple way to convert a state transition diagram
to Verilog is to write one conditional expression for
each transition in the diagram. This expression in-
cludes the starting state and the transition condition.
Only one of these logical expressions will be true and
so they can be written in any order.
If an expression is true then the state is set to the

ending state for that transition. If none of the condi-
tions match then the state is set to its current value
and so it doesn’t change.
// 2−bit counter with reset

module ex53
( output logic [1:0] count,
input logic reset, clk ) ;

always_ff @(posedge clk) count

<= count == 2'b00 && !reset ? 2'b01 :
count == 2'b01 && !reset ? 2'b10 :
count == 2'b10 && !reset ? 2'b11 :
count == 2'b11 ? 2'b00 :
count == 2'b01 && reset ? 2'b00 :
count == 2'b10 && reset ? 2'b00 : count ;

endmodule

Two conventions can simplify state transition ta-
bles: (1) An x (“don’t care”) can replace values that
don’t affect the next state. Each x halves the num-
ber of lines needed. (2) Arithmetic expressions can
replace lines where the next state can be computed
from the current state or the input.
We can write the state transition table for this

counter in 3 lines rather than 8 as:

count reset next
count

xx 1 00
11 0 00
𝑛 0 𝑛 + 1

and the Verilog would be:
// 2−bit counter with reset

module ex56
( output logic [1:0] count,
input logic reset, clk ) ;

always_ff @(posedge clk) count
<= reset ? 2'b00 :

count == 2'b11 ? 2'b00 :
count + 1'b1 ;

endmodule

Exercise 2: Simplify the solution using the fact that addition of 1 to
3 “wraps” a 2‑bit value to 0. What would you change so the counter
does not “wrap around” from 2’b11 to 2’b00?

0 1 ps 2 ps 3 ps 4 ps 5 ps 6 ps

xx 00 01 10

clk=0

count[1:0]=01

reset=0

Timers and Clock Dividers

We can create delays by counting clock cycles. It
takes 𝑁 clock periods for a counter to count down2
from 𝑁 − 1 to 0. If the clock period is 𝑇 then the de-
lay is 𝑁𝑇.
Exercise 3: What value of𝑁 would give a 20 ms delay if the clock
frequency is 50MHz? Howmanybits are needed for this timer’s reg‑
ister?

2Timers usually count down from𝑁 − 1 to 0 rather than up
from 0 to𝑁−1 because it’s simpler to determinewhen the count
has reached 0.

clk=0

count[4:0]=XXX

lights[5:0]=xxxxxx

reset=1

Figure 1: Simulation of traffic light controller.

If the counter is reset to 𝑁 − 1 when it reaches 0
then the count values will be periodic with a period
𝑁𝑇.
Exercise 4: Assume the timer above is reset to𝑁 − 1 each time
it reaches 0. For how long is the register value 0? What are the pe‑
riod and frequency of a signal that is inverted each time the count
reaches 0?

Traffic Light Controller

This is a controller for a traffic light at an intersection:

It combines two state machines: one to sequence the
traffic lights and one as a timer. The states are en-
coded as 6-bit values with the on/off values of the
(Red, Green, Yellow) lights in each direction:

Delays are implemented by decrementing a counter
on each clock edge. When the counter reaches zero
the state register is loaded with the next state and
the counter register is loaded with the duration of the
next state.
The state transition diagram for the light state ma-
chine is:

rg

gr

ryyr

count==0

count==0count==0

count==0

and for the timer:

29 28 4 3 0

state==ry || state==yr

state==rg || state==gr

A Verilog module implementing this state ma-
chine is:
// traffic light controller

module ex54
( output logic [5:0] lights,
input logic reset, clk ) ;

typedef enum logic [5:0] // RYG RYG
{ rg=6'b100_001, ry=6'b100_010,
gr=6'b001_100, yr=6'b010_100 }

lightstate ;

logic [4:0] count ;

// next traffic light lights
always @(posedge clk) lights

<= reset ? rg :
count ? lights :
lights == rg ? ry :
lights == ry ? gr :
lights == gr ? yr : rg ;

// state durations
always @(posedge clk) count

<= reset ? 29 :
count ? count-1 :
lights == rg || lights == gr ? 4 : 29 ;

endmodule

An enumerated type, lightstate, allows us to
use more meaningful symbolic names (rg, ry, gr,
and gy) for the four possible values of the state reg-
ister.
The simulation results are shown in Figure 1.

Exercise 5: Write the state transition table for the statemachine for
the lights output.

Sequence Detectors

This type of state machine detects particular se-
quences of input values. Examples include detecting
a change in an input, a specific sequence of input val-
ues or the duration of an input.
The most general implementation of a sequence

detector is a shift register that stores previous inputs.
Outputs are generated based on previous inputs – the
values in the shift register. However, a simpler imple-
mentation is often possible as shown in the examples
below.

Detecting Changes

An edge detector detects the change of an input be-
tween clock edges. The state is the input at the two
most recent clock edges. For example, 10means the
two most recent inputs were 1 followed by 0. The
rising output is true when the input changed from
low to high. The falling output is true for the op-
posite change. store the

00

01

11

10

1 0

1

0

1

in=0

Exercise 6: Forwhich states would falling be asserted? rising?
Draw the schematic andwrite the Verilog for this statemachine. As‑
sume an input in and a 2‑bit register bits that holds the twomost
recent input values.

Detecting Sequences of Values

In this example, a combination lock, the state is the
most recent four input digits. Combinational logic
asserts an unlock output when the most recent four
inputs match the passcode (1,2,3,4 in this example).
// digit−sequence detector

module ex24 ( output logic unlock,
input logic [3:0] digit,
input logic clk ) ;

logic [15:0] digits ;

// next−state logic

always_ff @(posedge clk) digits

<= { digits[11:0], digit } ;

// sequence detector output

assign unlock
= digits == { 4'd1, 4'd2, 4'd3, 4'd4 } ?
'1 : '0 ;

endmodule

0 1 us 2 us 3 us 4 us 5 us 6 us 7 us

x 5 1 2 3 4 2
xxxx xxx5 xx51 x512 5123 1234 2342

clk
digit[3:0]
digits[15:0]
unlock

Exercise 7: How could youmodify the code so that digits is only
updated when an enable input is asserted?
Exercise 8: Howmany states can this state machine have?
A simpler implementation would count the num-

ber of digits that had been entered in the correct or-
der.
Exercise 9: Draw the state transition diagram for this simpler im‑
plementation. How many states are there? Write the Verilog using
a 3‑bit count state variable.

Detecting Durations

Most mechanical switches briefly interrupt the con-
nection when they switch. A switch debouncer elim-
inates the extra edges on a switch input.
The debouncer below uses two state machines: a

timer to delay changing the output until the input has
been stable for𝑁 clock cycles and one to hold the cur-
rent output value until the timer expires. The timer
uses a register named count and the input and out-
put are named in and out respectively.

N-1 n-1 0

in!=out

in==out

in!=out

in==out

D Q out
0

1

clock

out

in

count == 0

Exercise 10: Write always_ff statements that implement these
two state machines.

0 1 us 2 us 3 us 4 us 5 us 6 us 7 us 8 us 9 us

80 40 20 10 08 04 02 01 00

xx 80 40 20 10 08 04 02 01 00

1234

XXX 128 0 64 32 48 40 36 34 35

clk=1

reset=0

delta_next[7:0]=00

delta[7:0]=00

done=1

n[15:0]=1234

sqrt[7:0]=35

Figure 2: Simulation of the calculation of square root of 1234.

RTL Controller

“Algorithmic state machines” control the computa-
tion of numerical results. A state machine, called the
the “controller,” controls the values loaded into reg-
isters, called the “datapath,” which hold numerical
values. This is called Register Transfer Level (RTL)
design.
This example computes the square root of an input

number by bisection.
The datapath contains two registers: one holds the

current best estimate of the square root, sqrt, and
the other the search interval, delta. On reset, sqrt
is loaded with the number whose square root it to be
computed and delta is set to half of the maximum
value. The controller state machine consists of a one-
bit register, done, which is reset to 0 and set to 1when
delta reaches 0.
Note that delta_next, the input to the delta reg-

ister, is a separate signal. This allows done to be set
to 1 at the same clock edge as delta is set to zero.
module ex41
(
input logic [15:0] n,
input logic reset, clk,
output logic [7:0] sqrt,
output logic done
) ;

logic [7:0] delta, delta_next ;

always_ff @(posedge clk) sqrt
<= reset ? 8'd128 :

{8'b0,sqrt} * sqrt < n ?
sqrt + delta : sqrt - delta ;

assign delta_next
= reset ? 8'd128 : delta/2 ;

always_ff @(posedge clk) delta
<= delta_next ;

always_ff @(posedge clk) done
<= delta_next == 0 ;

endmodule

Figure 2 shows the calculation of the square root

of 1234.
Exercise 11: What is the size of the expression sqrt*sqrt? Of
{8'b0,sqrt}*sqrt?
Exercise 12: Draw the state transition diagram for done.

One‑Hot State Encodings

“One-hot” state encodings use one flip-flop per state
and only one flip-flop at a time may be set to 1. This
requires more flip-flops but may require less combi-
national logic. For the first example (the motor con-
troller) the two states could be encoded in two differ-
ent ways:

state binary
encoding

one‑hot
encoding

off 0 10
on 1 01

Exercise13: Ɪfweused8‑bits of state information, howmany states
could be represented? What if we used 8 bits of state but used a
“one‑hot” encoding?

